







Subsidiary Recursion in Coq

Aaron Stump   

Computer Science Dept., The University of Iowa, USA

Alex Hubers 

Computer Science, The University of Iowa, USA

Christopher Jenkins  

Computer Science, The University of Iowa, USA

Benjamin Delaware  

Computer Science, Purdue University, USA

Abstract

This paper describes a functor-generic derivation in Coq of subsidiary recursion. With this recursion scheme, inner recursions may be initiated within outer ones, in such a way that outer recursive calls may be made on results from inner ones. The derivation utilizes a novel (necessarily weakened) form of positive-recursive types in Coq, dubbed retractive-positive recursive types. A corresponding form of induction is also supported. The method is demonstrated through several examples.

2012 ACM Subject Classification Software and its engineering → Recursion; Software and its engineering → Polymorphism

Keywords and phrases strong functional programming, recursion schemes, positive-recursive types, impredicativity

Digital Object Identifier 10.4230/LIPIcs.ITP.2022.

1 Introduction: subsidiary recursion

Central to interactive theorem provers like Coq, Agda, Isabelle/HOL, Lean and others are terminating recursive functions over user-declared inductive datatypes [8, 14, 17, 7]. Termination is usually enforced by a syntactic check for structural decrease, which is sufficient for many basic functions. For example, the `span` function from Haskell’s prelude (`Data.List`) takes a list and returns a pair of the maximal prefix whose elements satisfy a given predicate `p`, and the remaining suffix:

```
span :: (a -> Bool) -> [a] -> ([a], [a])
span _ []      = ([], [])
span p (x:xs) = if p x
                  then let (ys,zs) = span p xs in (x:ys,zs)
                  else ([], x:xs)
```

The sole recursive call is `span p xs`, and it occurs in a clause where the input list is of the form `x:xs`. Hence it is structurally decreasing. In the appropriate syntax, this definition can be accepted without additional effort by all the mentioned provers.

This paper is about a more expressive form of terminating recursion, called **subsidiary recursion**. While performing an outer recursion on some input `x`, one may initiate an inner recursion on `x` (or possibly some of its subdata), preserving the possibility of further invocations of the outer recursive function. Let us see a simple example. The function `wordsBy` (`Data.List.Extra`) breaks a list into its maximal sublists whose elements do not satisfy a predicate `p`. For example, `wordsBy isSpace " good day "` returns `["good","day"]`. Code is in Figure 1. Recall that `break p` is equivalent to `span (not . p)`. The first recursive call, `wordsBy p tl`, is structural. But in the second, we invoke `wordsBy p` on a value obtained



© Aaron Stump, Alex Hubers, Christopher Jenkins, and Benjamin Delaware;
licensed under Creative Commons License CC-BY 4.0

Interactive Theorem Proving 2022.

Editors: June Andronick and Leonardo da Moura; Article No. ; pp. 1–18

Leibniz International Proceedings in Informatics



LIPICs Schloss Dagstuhl – Leibniz-Zentrum für Informatik, Dagstuhl Publishing, Germany

XX:2 Subsidiary Recursion in Coq

```
wordsBy :: (a -> Bool) -> [a] -> [[a]]
wordsBy p [] = []
wordsBy p (hd:tl) =
  if p hd
  then wordsBy p tl
  else let (w,z) = break p tl in
        (hd:w) : wordsBy p z
```

■ **Figure 1** Haskell code for `wordsBy`, demonstrating subsidiary recursion

39 from another recursion, namely `span`. This is not allowed under structural termination, but
40 will be permitted by subsidiary recursion.

41 1.1 Summary of results

42 This paper presents a functor-generic derivation of terminating subsidiary recursion and
43 induction in Coq. We emphasize that this is a derivation within the type theory of Coq,
44 and requires no axioms or other modifications to Coq, except the `-impredicative-set` flag.
45 Using this derivation, we present several example functions like `wordsBy`, and prove theorems
46 about them. A nice example is a definition of run-length encoding using `span` as a subsidiary
47 recursion, where we prove that encoding and then decoding returns the original list. Our
48 approach applies to the standard datatypes in the Coq library, and does not require switching
49 libraries or datatype definitions.

50 An important technical novelty is a derivation of a weakened form of positive-recursive
51 type in Coq. Coq (Agda, and Lean) restrict datatypes D to be strictly positive: in the input
52 types of constructors of D , D cannot occur to the left of any arrows. Our derivation needs
53 to use positive-recursive types, where D may occur to the left of an even number (only)
54 of arrows. We present a way to derive a weakened form of positive-recursive type that is
55 sufficient for our examples (Section 4.1). The weakening is to require only that $F(\mu F)$ is a
56 retract of μF . Usually, these types are isomorphic. Hence, we dub these **retractive-positive**
57 recursive types. This weakening leads to noncanonical elements of μ , but we will see how to
58 work around this. Our definition of retractive-positive recursive types makes essential use of
59 impredicative quantification, and hence is not legal in predicative theories like Agda’s.

60 We begin by summarizing the interface our derivation provides for subsidiary recursion
61 (Section 2), and then see examples (Section 3). We next explain how the interface is actually
62 implemented (Section 4), including our retractive-positive recursive types (Section 4.1). The
63 interface for subsidiary induction is covered next (Section 5), and example proofs using it
64 (Section 6). Related work is discussed in Section 7.

65 All presented derivations have been checked with Coq version 8.13.2. The code may be
66 found as release `itp-2022` (dated prior to the ITP 2022 deadline) at <https://github.com/astump/coq-subsidiary>. The paper references files in this codebase, as an aid to the reader
67 wishing to peruse the code.

69 2 Interface for subsidiary recursion

70 This section presents the interface our Coq development provides for subsidiary recursion.

```

Definition List := Subrec ListF.
Definition inList : ListF List -> List := inn ListF.
Definition mkNil : List := inList Nil.
Definition mkCons (hd : A) (tl : List) : List := inList (Cons hd tl).
Definition toList : list A -> List.
Definition fromList : List -> list A.

```

■ **Figure 2** Some basics from `List.v`, specializing the functor-generic derivation of subsidiary recursion to lists parametrized by an element type `A` (`List.v`)

71 2.1 The recursion universe

72 Our approach is within a long line of work using ideas from universal algebra and category
 73 theory to describe inductive datatypes and their recursion principles (cf. [22, 5, 11]). With
 74 this approach, one describes transformations to be performed on data as *algebras*, which can
 75 then be *folded* over data. The simplest form of algebras, namely F -algebras for functor F
 76 (called the *signature functor* of the datatype), are morphisms from $F A$ to A , for carrier
 77 object A . From a programming perspective, an F -algebra is given input of type $F A$, and
 78 must compute a result of type A . An example of F is the signature functor for lists, which
 79 we will use below:

```

Inductive ListF(X : Set) : Set :=
| Nil : ListF X
| Cons : A -> X -> ListF X.

```

80 Algebras for our subsidiary recursion are more complex than F -algebras. Let us begin
 81 with an informal explanation. For reasons we will explain further below, the carrier of the
 82 algebra will be a functor $X : \text{Set} \rightarrow \text{Set}$. The algebra is presented with:

- 83 ■ a type $R : \text{Set}$, which will be this recursion’s view of the datatype.
- 84 ■ a function `fold` : `FoldT Alg R`, which allows one to initiate subsidiary recursions over
 85 data of type R . We will present the type `FoldT Alg R` below.
- 86 ■ a function `rec` : $R \rightarrow X R$, to use for making recursive calls, on any value of type R .
- 87 ■ and a *subdata structure* $d : F R$, where F is the signature functor for the datatype.

88 The algebra is then required to produce a value of type $X R$.

89 We will use Coq inductive types for the signature functors F of various datatypes. This
 90 allows algebras to use Coq’s pattern-matching on the subdata structure d . So the style
 91 of coding against this interface retains a similar feel to structural recursion. Unlike with
 92 structural termination, though, the interface here is type-based and hence compositional.

93 We have previously dubbed this interface a *recursion universe* [20]. As in other domains
 94 using the term “universe”, we have a kind of space (here, R), which one cannot escape using
 95 certain operations. Other examples are the ordinal ϵ_0 and ω^- , and the physical universe and
 96 traveling at the speed of light. Staying in the recursion universe is good, because we may
 97 recurse (via `rec`) on any value of type R . Some points must still be explained: why X has
 98 type $\text{Set} \rightarrow \text{Set}$, and the definition of `FoldT`. Let us see these details next.

99 2.2 Types for subsidiary recursion (`Subrec.v`, `List.v`)

100 The type over which one can recurse using our scheme of subsidiary recursion is called
 101 `Subrec`. It is parametrized by a signature functor F of type $\text{Set} \rightarrow \text{Set}$. `Subrec` comes with

XX:4 Subsidiary Recursion in Coq

```
Definition KAlg : Type := (Set -> Set) -> Set.

Definition FoldT(alg : KAlg)(C : Set) : Set :=
  forall (X : Set -> Set) (FunX : Functor X), alg C X -> C -> X C.

Definition AlgF(Alg: KAlg)(X : Set -> Set) : Set :=
  forall (R : Set)
    (fold : FoldT Alg R)
    (rec : R -> X R)
    (d : F R),
    X R.

Definition Alg : KAlg := MuAlg AlgF.

Definition fold : FoldT Alg Subrec.
Definition rollAlg : forall {X : Set -> Set}, AlgF Alg X -> Alg X.
Definition unrollAlg : forall {X : Set -> Set}, Alg X -> AlgF Alg X.
```

■ **Figure 3** The type for algebras, parametrized over $F : \text{Set} \rightarrow \text{Set}$ (`Subrec.v`)

102 `inn : F Subrec -> Subrec`, which behaves computationally like a constructor. We will
103 later derive an induction principle for this type (Section 5). The definition of `Subrec` uses
104 retractive-positive recursive types, to take a fixed-point of a construction based on `F`. We
105 present these recursive types in Section 4.1 below.

106 In this paper, we use just the specialization to the case of lists, with signature functor
107 `ListF A` shown above. The parameter `A` is the type for the list elements. `List` is then
108 defined to be `Subrec`, with `F` instantiated to `ListF A`. In general, to use our development
109 to get subsidiary recursion over some datatype, one must define a signature functor for
110 the datatype. Note that `List` is different from the type `list` of lists in Coq's standard
111 library. Our development is meant to be used in extension of existing inductive datatypes,
112 not replacing them. The figure also shows constructors `mkNil` and `mkCons` for `List`, and
113 typings for conversion functions between `List` and `list` (definitions elided).

114 2.3 Algebras for subsidiary recursion

115 `Subrec.v` also implements the notion of algebra we introduced informally above. The central
116 definitions are in Figure 3. `KAlg` is the kind for the type-constructor for algebras, as we see
117 in the definition of `Alg`. This type-constructor `Alg` is a fixed-point of the type `AlgF`. The
118 fixed-point is taken using `MuAlg`, which implements our retractive-positive recursive types
119 (Section 4.1) at kind `KAlg`.

120 We need a fixed-point here because `Alg` occurs in the definition of `AlgF`. This is an
121 essential circularity, because we are trying to express that algebras take in `fold` functions,
122 which themselves may accept algebras. The variable `Alg` occurs negatively in `FoldT Alg R`
123 which occurs negatively in `AlgF Alg X`. Hence it occurs positively but not strictly positively.
124 So we can indeed take a fixed-point of `AlgF` to define the constant `Alg`.

125 Let us look at `AlgF`. As noted already, each recursion is based on an abstract type `R`,
126 representing the data upon which we will recurse. This is the first argument to a value of
127 type `AlgF Alg X`. Reasoning parametrically, an algebra can assume nothing about `R` except

```

Theorem FoldChar :
  forall (X : Set -> Set) (FunX : Functor X) (IdF : FmapId X FunX)
    (algf : AlgF Alg X) (d : F Subrec),
  fold X FunX (rollAlg algf) (inn d) =
    algf _ fold (fold X FunX (rollAlg algf)) d .

```

■ **Figure 4** Computation law for subsidiary recursion, stated as a theorem

128 that it supports the following operations. We have a local `fold` function, which will allow us
 129 to fold another algebra over data of type `R`. We will use `fold` to initiate subsidiary recursions.
 130 Then there is `rec`, for recursive calls on data of type `R`.

131 As noted already, for subsidiary recursion, algebras have a carrier `X` which depends
 132 (functorially) on a type. When we fold an algebra using a fold function (either global or local)
 133 of type `FoldT Alg C`, (i) recursive calls may compute a result of type `X R`, mentioning the
 134 abstract type `R` for that recursion; and (ii) outside that recursion, the result will have type
 135 `X C`. Having a functor for the carrier of the algebra gives us the flexibility to type results
 136 inside a recursion with the abstract type `R`, but view those results as having the type `C`
 137 outside the recursion. The function `fold` (Figure 3) initiates top-level folds. We also can
 138 have functions between `Alg` and its `Algf`-unfolding. We will return to the code for `Subrec.v`
 139 in Section 4.

140 Finally, for a recursion scheme, one would like to see not just the typed interface, but
 141 also the computation law. This is shown as a theorem in Figure 4. Intuitively, it states that
 142 folding an algebra over constructed data `inn d` is equal to invoking the algebra on `fold` for
 143 the fold function; an invocation of `fold` with the algebra for the `rec` function; and `d` for the
 144 subdata structure.

145 3 Examples of subsidiary recursion

146 Having seen the interface for subsidiary recursion in Coq, let us consider now some examples.

147 3.1 The span function (`Span.v`)

148 This first example does not invoke subsidiary recursions, but will itself be used as a subsidiary
 149 recursion in other examples to follow. Given a predicate `p : A -> bool`, and a value of
 150 type `List A`, we would like to compute a pair of type `list A * List A`, where the first
 151 component is the maximal prefix whose elements satisfy `p`, and the second is the remaining
 152 suffix. This is the typing for a top-level recursion. More generally, though, given a type
 153 `R : Set` along with a fold function for that type (i.e., of type `FoldT (Alg (ListF A)) R`),
 154 we will map an input list of type `R` to a pair of type `list A * R`. The first component of this
 155 pair is going to be built up from scratch, and so cannot have type `R`; we cannot statically
 156 ensure that outer recursions on it are legal. But the second component will always be a
 157 subdatum of the input list, and so can still have type `R`, enabling outer recursive calls. So we
 158 want:

```

Definition spanr{R : Set}(fo:FoldT (Alg (ListF A)) R)
  (p : A -> bool)(xs : R) : list A * R.

```

159 From this we can also define the top-level recursion, by supplying `fold (ListF A)`, which is
 160 the function for folding an algebra over a list (Figure 3), for the argument `fo` of `spanr`:

XX:6 Subsidiary Recursion in Coq

```
Definition SpanAlg(p : A -> bool) : Alg (ListF A) SpanF :=
  rollAlg (fun R fo span xs =>
    match xs with
    | Nil => SpanNoMatch
    | Cons hd tl =>
      if p hd then
        match (span tl) with
        | SpanNoMatch => SpanSomeMatch [hd] tl
        | SpanSomeMatch l r => SpanSomeMatch (hd::l) r
        end
      else
        SpanNoMatch
    end).
end).
```

■ **Figure 5** The algebra `SpanAlg` for the `span` function (`Span.v`)

```
Definition span(p : A -> bool)(xs : List A) : list A * List A
:= spanr (fold (ListF A)) p xs.
```

161 Before we define `spanr`, we must resolve a small problem. If the first element of the input
162 list `xs` to `span` does not satisfy `p`, then `span` should return `([], xs)`. But when recursing
163 on `xs`, we will see it only in the form of a subdata structure of type `ListF A R`. We will not
164 be able to return it from our recursion at type `R`, and hence we would not be able to return
165 `([], xs)` as desired. To work around this, we will have our recursion return a value of type
166 `SpanF R` (`X` will be implicit for the constructors):

```
Inductive SpanF(X : Set) : Set :=
  SpanNoMatch : SpanF X
| SpanSomeMatch : list A -> X -> SpanF X.
```

167 The idea is that the recursion will return `SpanNoMatch` to signal that it is in the one tricky
168 case where `p` does not match the first element. Otherwise, it will be able to return, via
169 `SpanSomeMatch`, a prefix and the suffix at type `R`. The prefix will be nonempty, and hence
170 the suffix will be at most the tail of `xs`. This suffix is available to the algebra in the subdata
171 structure of type `ListF A R`.

172 3.1.1 The algebra for `span`

173 Figure 5 shows the algebra `SpanAlg`, whose type is `Alg (ListF A) SpanF`. So we are
174 defining an algebra (`Alg`) for the `ListF A` functor, with carrier `SpanF` of the required
175 type `Set -> Set`. We use `rollAlg` to create an algebra from something whose type is an
176 application of `AlgF`. This takes in all the components of the recursion universe: the abstract
177 type `R`, the fold function (`fo`) for any subsidiary recursions (not needed here), a function we
178 choose to name `span` for making recursive calls, and finally `xs : ListF A R`. The algebra
179 pattern-matches on this `xs`. In the cases where it is empty or where its head (`hd`) does not
180 satisfy `p`, we return `SpanNoMatch`. This signals to the caller that we really wished to return
181 `([], xs)`, but could not because we do not have `xs` at type `R`. If the head does satisfy `p`, then
182 we recurse on the tail (`tl : R`) by calling the provided `span : R -> SpanF R`. If `span tl`
183 returns `SpanNoMatch`, that means that we should make `tl` the suffix in the pair we return

```

Definition spanhr{R : Set}(fo:FoldT (Alg (ListF A)) R)
  (p : A -> bool)(xs : R) : SpanF R :=
  fo SpanF SpanFunctor (SpanAlg p) xs.

Definition spanr{R : Set}(fo:FoldT (Alg (ListF A)) R)
  (p : A -> bool)(xs : R) : list A * R
:= match spanhr fo p xs with
  SpanNoMatch => ([],xs)
  | SpanSomeMatch l r => (l,r)
end.

Definition breakr{R : Set}(fo:FoldT (Alg (ListF A)) R)
  (p : A -> bool)(xs : R) : list A * R :=
  spanr fo (fun x => negb (p x)) xs.

```

■ **Figure 6** Functions derived from SpanAlg (Span.v)

184 (via SpanSomeMatch). Happily, we have `tl : R` here, so we can do this. For either possible
 185 return value of `span tl`, we add the head to the front of the prefix.

186 3.1.2 Defining span from SpanAlg

187 SpanAlg is used in the definition of `spanhr`, in Figure 6. This function invokes the fold
 188 function it is given, on SpanAlg. The final twist is now in the definition of `spanr`. We call
 189 `spanhr` on the input `xs : R`. If `spanhr` returns `SpanNoMatch`, then we are supposed to return
 190 `([],xs)`, which we can do here, because we have `xs : R`. It was only inside the algebra that
 191 we lost the information that the subdata structure of type `F R` is derived from a value of
 192 type `R`. If `spanhr` returns `SpanSomeMatch l r`, then we return the nonempty prefix `(l)` and
 193 the suffix `(r)`. We also define a version of `break` for subsidiary recursion (e.g., in `wordsBy`,
 194 below).

195 3.2 The wordsBy function (WordsBy.v)

196 We now consider how to write the `wordsBy` function from Section 1, using `breakr` subsidiarily.
 197 The code is in Figure 7, assuming a type `A : Set`. The setup is similar to that for `span`.
 198 We first define an algebra `WordsByAlg` of type `Alg (ListF A) (Const (list (list A)))`,
 199 parametrized by a predicate `p`. This type expresses that `WordsByAlg p` is an algebra (Alg)
 200 for the `ListF A` functor, with carrier `Const (list (list A))`. `Const` is a combinator for
 201 creating the object part of constant functors; `FunConst` creates the morphism part (i.e., the
 202 `fmap` function). We use `Const` where the return type of the algebra will not depend on its
 203 abstract type `R`. Since we are constructing a list of lists from scratch, it will not be legal
 204 to recurse on the list itself, or its (list) elements. So we just use the `list` type of Coq's
 205 standard library.

206 The code for `WordsByAlg` is essentially the same as what we saw in Section 1. Recall that
 207 this function drops elements that satisfy `p`, and returns the list of sublists between maximal se-
 208 quences of such elements. The algebra pattern-matches on `xs : ListF A R`. In the `Cons` case,
 209 if the head (`hd`) satisfies the predicate, then we drop it and recurse. Legality of the recursive
 210 call follows by typing: `tl : R` has the type expected by `wordsBy : R -> list (list A)`.

XX:8 Subsidiary Recursion in Coq

```
Definition WordsByAlg(p : A -> bool)
  : Alg (ListF A) (Const (list (list A))) :=
  rollAlg (fun R fo wordsBy xs =>
    match xs with
    | Nil => []
    | Cons hd tl =>
      if p hd then
        wordsBy tl
      else
        let (w,z) := breakr fo p tl in
        (hd :: w) :: wordsBy z
    end).
Definition wordsByr{R : Set}(fo:FoldT (Alg (ListF A)) R)
  (p : A -> bool)(xs : R) : list (list A) :=
  fo (Const (list (list A))) (FunConst (list (list A))) (WordsByAlg p) xs.
```

■ **Figure 7** Functions `wordsBy` and `wordsByr`, and the algebra they fold (`WordsBy.v`)

```
mapThrough :: (a -> [a] -> (b, [a])) -> [a] -> [b]
mapThrough f [] = []
mapThrough f (a:as) = b : mapThrough f as'
  where (b, as') = f a as
```

■ **Figure 8** The `mapThrough` function in Haskell

211 Otherwise, we use `breakr` to obtain the maximal prefix `w` of `tl` that does not satisfy `p`, and
212 the remaining suffix `z`.

213 Here we see the benefit of our approach. From Figure 6, the return type of `breakr`
214 is `list A * R`, where `R` comes from the type `FoldT (ListF A) Alg R` of `fo`, from the
215 definition of `AlgF` in Figure 3 (instantiating the functor with `ListF A`). This means that
216 from the invocation of `breakr`, we get `w : list A` and `z : R`. Thus, it is legal to apply
217 `wordsBy : R -> list (list A)` to `z` to recurse. The figure also shows the code for the
218 subsidiary recursion `wordsByr`.

219 3.3 The `mapThrough` function (`MapThrough.v`)

220 This example shows how to write a combinator that factors out a subsidiary recursion.
221 The Haskell library `Data.List.Extra` defines a function `repeatedly` in essentially the same
222 way as `mapThrough` in Figure 8 (we propose a more informative name). This function behaves
223 like the standard `map` function on lists, except that the function `f` that we are mapping (or
224 “mapping through”) takes in not just the current element `a`, but also the tail `as`. It then
225 returns the value `b` to include in the output list, and whatever other list it wishes, upon
226 which `mapThrough` will then recurse.

227 To write this combinator using our infrastructure for subsidiary recursion, we need to
228 supply the mapped function with the fold function for `mapThrough`’s recursion. This is so
229 that the mapped function can initiate a subsidiary recursion, returning a value in the abstract
230 type `R` of `mapThrough`’s recursion. The type we will use for mapped functions is:

```
Definition mappedT(A B : Set) : Set :=
```



```

Definition MapThroughAlg{B : Set}(f:mappedT A B)
  : Alg (ListF A) (Const (list B)) :=
  rollAlg (fun R fo mapThrough xs =>
    match xs with
    | Nil => []
    | Cons hd tl =>
      let (b,c) := f R fo hd tl in
      b :: mapThrough c
    end).
Definition mapThroughr{R : Set}(fo:FoldT (Alg (ListF A)) R)
  {B : Set}(f:mappedT A B) : R -> list B.
Definition mapThrough{B : Set}(f:mappedT A B) : List A -> list B.

```

■ **Figure 9** The algebra MapThroughAlg defining the functions mapThrough and mapThroughr; code for the latter is omitted, as it follows the pattern of wordsBy and wordsByr of Figure 7 (MapThrough.v)

```

rle :: Eq a => [a] -> [(Int,a)]
rle = mapThrough compressSpan
  where compressSpan a as =
    let (p,s) = span (== a) as in
    ((1 + length p, a),s)

```

■ **Figure 10** Run-length encoding in Haskell, using mapThrough and span

```

forall(R : Set)(fo:FoldT (Alg (ListF A)) R), A -> R -> B * R.

```

231 This type is more informative than the Haskell type, since it shows that the second component
 232 of the returned value must have type R, and hence must be (hereditarily) a tail of the input.

233 Given this definition, the Coq definition of mapThrough is shown in Figure 9. MapThroughAlg
 234 is similar to the Haskell code above, though when we call f, we must supply the abstract
 235 type R and fold function fo. From the definition of mappedT, we have that b : B and c : R,
 236 so we may indeed invoke mapThrough : R -> list B on c. Note that as we are building
 237 up a new list from scratch (rather than just extracting some tail of the input list), we just
 238 return list B; we cannot perform further subsidiary recursion on the output.

239 3.4 Run-length encoding (Rle.v)

240 Finally, we have an example using our mapThrough combinator together with a subsidiary
 241 recursion, to implement *run-length encoding*. This is a basic data-compression algorithm
 242 where maximal sequences of n occurrences of element e are summarized by the pair (n, e) [19].
 243 A Haskell implementation of this algorithm is in Figure 10. Recall that $(== a)$ tests its
 244 input for equality with a . The compressSpan helper function gathers up all elements at
 245 the start of the tail as that are equal to the head a . This prefix is returned as p , with the
 246 remaining suffix as s . The pair $(1 + \text{length } p, a)$ is returned to summarize $a :: p$. The
 247 mapThrough combinator then iterates compressSpan through the suffix s .

248 Assuming $A : \text{Set}$ and an equality test $\text{eqb} : A \rightarrow A \rightarrow \text{bool}$ on it, we port this code
 249 to our Coq infrastructure in Figure 11. The function compressSpan is written at the type
 250 mappedT A (nat * A) that will be required by mapThrough. Unfolding the definition of
 251 mappedT, we see that compressSpan has this type:

XX:10 Subsidiary Recursion in Coq

```
Definition compressSpan : mappedT A (nat * A) :=
  fun R fo hd tl =>
    let (p,s) := spanr fo (eqb hd) tl in
    ((succ (length p),hd), s).

Definition RleCarr := Const (list (nat * A)).
Definition RleAlg : Alg (ListF A) RleCarr :=
  MapThroughAlg compressSpan.
Definition rle(xs : List A) : list (nat * A)
  := fold (ListF A) RleCarr (FunConst (list (nat * A))) RleAlg xs.
```

■ **Figure 11** The function `rle` for run-length encoding, and the algebra `RleAlg` defining it in terms of `MapThroughAlg` of Figure 9 (`Rle.v`)

```
forall(R : Set)(fo:FoldT (Alg (ListF A)) R), A -> R -> (nat * A) * R.
```

252 It is invoked by the code for `mapThrough` with `fo : FoldT (Alg (ListF A)) R`. Then
253 `compressSpan` will extract from the tail at type `R` (second input) the suffix upon which
254 `mapThrough` should recurse (second component of the output pair). Then we define an algebra
255 `RleAlg` by supplying `compressSpan` as the function to map through, to `MapThroughAlg`
256 (Figure 9). Following the pattern seen above, we define function `rle` for top-level recursions
257 using `fold` (we could also define a subsidiary version `rlerr`).

258 4 Derivation of subsidiary recursion

259 Let us now consider the implementation of the interface we have used for the preceding
260 examples. The first step is our weakened form of positive-recursive types.

261 4.1 Retractive-positive recursive types (`Mu.v`)

262 As we have seen, our definitions require a form of positive-recursive types, to allow algebras
263 to accept fold functions that themselves require algebras, and also for the definition of
264 `Subrec` (which we will see in more detail in the next section). Full positive-recursive types
265 are incompatible with Coq's type theory [6]. One can impose some restrictions on large
266 eliminations which then enable positive-recursive types [3], but this requires changing the
267 underlying theory. Here we exploit Coq's impredicative polymorphism to obtain a different
268 solution.

269 Our starting point is a type scheme `F : Set -> Set`, with an `fmap` function (morphism
270 part of the functor) of type

```
forall A B : Set, (A -> B) -> F A -> F B
```

271 which satisfies the identity-preservation law for functors:

```
fmapId : forall (A : Set)(d : F A), fmap (fun x => x) d = d
```

272 Then we make the definitions of Figure 12. The critical idea is embodied in the definition of
273 `Mu`. Ideally, we would like to have a definition like

```
Inductive Mu' : Set := mu' : F Mu' -> Mu'.
```

```

Inductive Mu : Set :=
  mu : forall (R : Set), (R -> Mu) -> F R -> Mu.

Definition inMu(d : F Mu) : Mu :=
  mu Mu (fun x => x) d.

Definition outMu(m : Mu) : F Mu :=
  match m with
  | mu A r d => fmap r d
  end.

Lemma outIn(d : F Mu) : outMu (inMu d) = d.

```

■ **Figure 12** Derivation of retractive-positive recursive types (Mu.v)

274 This is exactly what is used in many approaches to modular datatypes in functional program-
 275 ming, like Swierstra’s [21]. But this definition is (rightly) rejected by Coq, as instantiations
 276 of F that are not strictly positive would be unsound.

277 Instead, we define Mu in Figure 12, to weaken this ideal Mu' to a strictly positive
 278 approximation. Instead of taking in $F \text{ Mu}$, the constructor mu accepts an input of type
 279 $F R$, for some type R for which we have a function of type $R \rightarrow \text{Mu}$. The impredicative
 280 quantification of R is essential here: we will instantiate it with Mu itself in the definition
 281 of inMu (Figure 12). So this approach would not work in a predicative theory like Agda’s.
 282 The quantification of R can be seen as applying a technique due to Mendler, of introducing
 283 universally quantified variables for problematic type occurrences, to a datatype constructor.
 284 We will review this in Section 7.

285 Returning to Figure 12, we have functions inMu and outMu , which make $F \text{ Mu}$ a retraction
 286 (outIn) of Mu : the composition of outMu and inMu is (extensionally) the identity on $F \text{ Mu}$.
 287 But the reverse composition cannot be proved to be the identity, because of the basic problem
 288 of **noncanonicity** that arises with this definition.

289 For a simple example: suppose we instantiate F with $\text{ListF } A$ (from Section 2.1). Our
 290 derivation uses a different type that wraps F , but using $\text{ListF } A$ demonstrates the issue in
 291 a simple form. Let us temporarily define $\text{List } A$ as $\text{Mu } (\text{ListF } A)$ (again, for subsidiary
 292 recursion do not use just ListF directly). The canonical way to define the empty list would
 293 be:

```

Definition mkNil := mu (List A) (fun x => x) (NilF A)

```

294 But given this, there are infinitely many other equivalent definitions. For any $Q : \text{Set}$, we
 295 could take

```

Definition mkNil' := mu Q (fun x => mkNil) (NilF A)

```

296 Since $\text{fmap } f \text{ (NilF } A)$ equals $\text{NilF } B$ for $f : A \rightarrow B$, if we apply outMu (of Figure 12) to
 297 mkNil' or mkNil , we will get $\text{NilF } (\text{List } A)$. But critically, mkNil and mkNil' are not equal,
 298 neither definitionally nor provably. Of course, one could define a function that puts Mu values
 299 in canonical form by folding inMu over them. Then mkNil and mkNil' would be equivalent.
 300 But they would still not be provably equal, which is the problem of noncanonicity. We will
 301 see how to work around this in Section 6. First, though, let us complete the exposition of
 302 our implementation of subsidiary recursion.

XX:12 Subsidiary Recursion in Coq

```
Definition SubrecF(C : Set) :=  
  forall (X : Set -> Set) (FunX : Functor X), Alg X -> X C.  
Definition Subrec := Mu SubrecF.  
Definition roll: SubrecF Subrec -> Subrec.  
Definition unroll: Subrec -> SubrecF Subrec.
```

■ **Figure 13** Definition of `Subrec` as a fixed-point of `SubrecF` (`Subrec.v`)

303 4.2 The implementation of `Subrec` (`Subrec.v`)

304 The type `Subrec` is defined in Figure 13, as a fixed-point of `SubrecF : Set -> Set`. We
305 build this fixed-point using `Mu` from the previous section, and obtain `roll` and `unroll`
306 functions between `SubrecF Subrec` and `Subrec`. Unrolling a `Subrec` term gives us a term
307 of type

```
forall (X : Set -> Set) (FunX : Functor X), Alg X -> X Subrec
```

308 So we see that `Subrec` is the type of functions which, for all algebras with functorial carrier
309 `X`, compute a value of type `X Subrec`. This is a generalization of the functor-generic type
310 $\forall X. Alg X \rightarrow X$ for the Church encoding, where $Alg X$ is $F X \rightarrow X$. We elide the
311 implementation of the `roll` and `unroll` functions, but we note that `unroll` makes use of
312 functoriality of carriers `X`.

313 The rest of the interface for `Subrec` is shown in Figure 14. To fold an algebra `alg` with
314 carrier `X` (with `fmap` function given by `FunX`) over `d : Subrec`, we `unroll` the definition of
315 `Subrec` and apply that to the algebra (with its carrier).

316 More interesting is the definition of `inn`, which is the critical point where the recursion
317 universe is implemented. To create a value of type `Subrec` from data of type `F Subrec`, the
318 definition of `inn` rolls a value of type `SubrecF Subrec` (we saw this type unfolded at the
319 start of this section). This value takes in a carrier `X`, its `fmap` function `xmap`, and an algebra
320 `alg` with that carrier. It will then call `alg` (after unrolling it) with implementations for the
321 components of the recursion universe (cf. Section 2.1, also Figure 3):

- 322 ■ `Subrec` is passed as the value for the abstract type `R`; this is what enables all the rest of
323 the components to have the desired types, since we will pass values that have `Subrec`
324 where the interface mentions `R`.
- 325 ■ The function `fold : FoldT Alg Subrec` is passed as the fold function of type `FoldT Alg R`.
- 326 ■ For the `rec : R -> X R` function, we pass `(fold X xmap alg) : Subrec -> X Subrec`.
- 327 ■ For the subdata structure of type `F R`, we pass `d : F Subrec`.

328 Finally, Figure 14 defines `out` as a subsidiary recursion, given a fold function. Outside the
329 recursion, `d` has type `F R`; inside the recursion it has type `F R'` where `R'` is the abstract type
330 of the subsidiary recursion. Intuitively, `out` implements the idea that unfolding an abstract
331 type one step is just a trivial case of subsidiary recursion.

332 5 Interface for subsidiary induction (`Subreci.v`)

333 We have seen how to write subsidiary recursions in Coq. But can one reason about these? We
334 turn now briefly to the interface to our development of subsidiary induction in Coq, and some
335 example proofs written using this interface. Subsidiary induction is the natural extension
336 of subsidiary recursion, which worked over `Sets`, to `Subrec`-predicates. The development is
337 parametrized by a functor `F` and a functor `Fi : (Subrec -> Prop) -> (Subrec -> Prop)`

```

Definition fold : FoldT Alg Subrec :=
  fun X FunX alg d => unroll d X FunX alg.

Definition inn : F Subrec -> Subrec :=
  fun d => roll (fun X xmap alg =>
    unrollAlg alg Subrec fold (fold X xmap alg) d).

Definition out{R:Set}(fo:FoldT Alg R) : R -> F R :=
  fo F FunF (rollAlg (fun R' _ _ d => d)).

```

■ **Figure 14** The rest of the interface for Subrec (Subrec.v)

338 over Subrec-indexed propositions (i.e., predicates). Just as functors need an `fmap` function,
 339 here we need an indexed version, of type `fmapiT Subrec Fi` (definition elided.)

340 The central definitions for the type `Subreci : Subrec -> Prop` are given in Figure 15.
 341 Where having a value `x` of `Subrec` entitles us to define subsidiary recursions to inhabit types
 342 `X Subrec`, a value of type `Subreci x` lets us prove properties of `x` by subsidiary induction.
 343 Briefly: `kMo` is the kind for *motives*, namely predicates on `Subrec` [15]. `KAlgi` is the kind
 344 for indexed algebras. `FoldTi` is the indexed version of `FoldT`: it expresses provability of `X C`
 345 for `d`, based on an indexed algebra and a value of type `C d`, where `C` is the (indexed) anchor
 346 type. `AlgFi` and `Algi` are indexed versions of the algebras we saw for recursion. The `rec`
 347 function from Figure 3 is now an induction hypothesis: given any `d` where `R d` holds, `ih`
 348 proves `X R d`. A value of type `R d` is thus a license to induct on `d`. Finally, the algebra
 349 is given a subdata structure indexed by `d : Subrec`, and must produce a proof of `X R d`.
 350 `Subreci` is defined as the suitably indexed fixed-point of `SubrecFi`, which is the natural
 351 indexed version of `SubrecF`.

352 For lists, we instantiate `Fi` with `ListFi`, shown in Figure 16. This is just the indexed
 353 version of `ListF`. Given a list `A`, `toListi` returns a value of type `Listi (toList xs)`.
 354 This can be understood as saying that for any list (from Coq’s standard library), we can
 355 reason by subsidiary induction to prove properties of `toList xs`. We also introduce an
 356 abbreviation `ListFoldTi` for the type of indexed fold functions over lists.

357 6 Examples of subsidiary induction

358 To prove the main theorem about run-length encoding, we need the three lemmas about
 359 `span` shown in Figure 17. For lack of space, we just state the properties. The first says that
 360 appending the results of a call to `span` returns the original list (module some conversions to
 361 `list` from `List`). The second uses the inductive proposition `Forall` from Coq’s standard
 362 library to state that all the elements of the prefix returned by `span` satisfy `p`. These lemmas
 363 are proved using indexed algebras with constant (indexed) carriers. In contrast, `GuardPresF`
 364 uses its argument `S` to express that whenever `spanh` returns a suffix `r`, that suffix satisfies
 365 `S`. This enables us to invoke an outer induction hypothesis on this suffix, when reasoning
 366 subsidiarily about `span`. Using these lemmas, we can write a short proof by subsidiary
 367 induction of the following theorem, where `rld : list (nat * A) -> list A` is the obvious
 368 decoding function:

```
Theorem RldRle (xs : list A): rld (rle (toList xs)) = xs.
```

XX:14 Subsidiary Recursion in Coq

```
Definition kMo := Subrec -> Prop.
Definition KAlgi := (kMo -> kMo) -> Set.
Definition FoldTi(alg : KAlgi)(C : kMo) : kMo :=
  fun d => forall (X : kMo -> kMo) (xmap : fmapiT Subrec X),
    alg X -> C d -> X C d.

Definition AlgFi(A : KAlgi)(X : kMo -> kMo) : Set :=
  forall (R : kMo)
    (fo : (forall (d : Subrec), FoldTi A R d))
    (ih : (forall (d : Subrec), R d -> X R d))
    (d : Subrec),
    Fi R d -> X R d.

Definition Algi := MuAlgi Subrec AlgFi.

Definition SubrecFi(C : kMo) : kMo :=
  fun d => forall (X : kMo -> kMo) (xmap : fmapiT Subrec X), Algi X -> X C d.
Definition Subreci := Mui Subrec SubrecFi.

Definition foldi(i : Subrec) : FoldTi Algi Subreci i.
Definition inni(i : Subrec)(fd : Fi Subreci i) : Subreci i.
```

■ **Figure 15** Interface for subsidiary induction (Subreci.v)

```
Definition lkMo := List -> Prop.

Inductive ListFi(R : lkMo) : lkMo :=
  nilFi : ListFi R mkNil
| consFi : forall (h : A)(t : List), R t -> ListFi R (mkCons h t).

Definition Listi := Subreci ListF ListFi.
Definition toListi(xs : list A) : Listi (toList xs) := listFoldi xs Listi inni.
Definition ListFoldTi(R : List -> Prop)(d : List) : Prop :=
  FoldTi ListF (Algi ListF ListFi) R d.
```

■ **Figure 16** The indexed version ListFi of ListF (List.v)

```

Definition SpanAppendF(p : A -> bool)(xs : List A) : Prop :=
  forall (l : list A)(r : List A) ,
    span p xs = (l,r) ->
    fromList xs = l ++ (fromList r).

```

```

Definition spanForallF(p : A -> bool)(xs : List A) : Prop :=
  forall (l : list A)(r : List A),
    span p xs = (l,r) ->
    Forall (fun a => p a = true) l.

```

```

Definition GuardPresF(p : A -> bool)(S : List A -> Prop)(xs : List A) : Prop :=
  forall (l : list A)(r : List A),
    spanh p xs = SpanSomeMatch l r ->
    S r.

```

■ **Figure 17** Statements of three lemmas about `span` (directory `SpanPfs`)

```

Definition spanForall2F(p : A -> bool)(xs : List A) : Prop :=
  Forall (fun a => p a = true) (fromList xs) ->
  span p xs = (fromList xs, getNil xs).

```

■ **Figure 18** A statement of the property that `span` returns the empty suffix, computed using `getNil` to avoid noncanonicity problems, if all elements satisfy `p`

369 We invoke the lemmas about `span` subsidiarily, so that we may apply our induction hypothesis
 370 to the suffix that `span` returns (on which `mapThrough` then recurses). For example, the
 371 lemma for `GuardPresF` takes in the indexed fold function `foi` from the outer induction (for
 372 `RldRle`), to show that the abstract predicate `R` applies to the suffix `r` returned by `span`. This
 373 enables the outer induction hypothesis (for `RldRle`) to be applied.

```

Lemma guardPres{R : List A -> Prop}(foi:forall d : List A, ListFoldTi R d)
  (p : A -> bool)(xs : List A)(rxs : R xs)
  (l:list A)(r : List A)(e: span p xs = (l,r)) : R r.

```

374 Finally, as promised, a note on noncanonicity. When proving properties about subsidiary
 375 recursions on `xs : List A`, one should be aware that nothing prevents the property from
 376 being applied to noncanonical `Lists`. For example, suppose we wish to prove that if all
 377 elements of a list satisfy `p`, then the suffix returned by `span` is empty. It is dangerous to
 378 phrase this as “the suffix equals `mkNil`”, because for a noncanonical input `xs`, `span` will
 379 return that same noncanonical `xs` as the suffix (and so it may be a noncanonical empty list,
 380 not equal to `mkNil`). The solution in this case is to use a function `getNil (List.v)` that
 381 computes an empty list from `xs`. The statement that one can prove is shown in Figure 18.

382 7 Related Work

383 **Termination.** In some tools, like Coq, Agda, and Lean, termination is checked statically,
 384 based on structural decrease. Others, like Isabelle/HOL, allow one to write recursions first,
 385 and prove (possibly with automated help) their termination afterwards [12]. These tools all
 386 support well-founded recursion, but in constructive type theory, evidence of well-foundedness

then propagates through code. In contrast, our approach here, while less general, does not clutter code with proofs. Subsidiary recursion can be seen as a generalization of *nested recursion*, which allows recursive calls of the form $f(f\ x)$ [13]. In subsidiary recursion, these are generalized to the form $f(g\ x)$, where g could be f or another recursively defined function. See the survey by Bove et al. for more on partiality and recursion in theorem provers [4].

Our work contributes to the program proposed by Owens and Slind, of broadening the scope of functional programs that can be accommodated in ITPs [18]. The goal of terminating recursion has been advocated in the literature on programming languages under the name *strong functional programming* [23]. Our method is similar to the technique of sized types, in providing a type-based method for termination [2]. With sized types, datatypes are indexed with abstract sizes, which must then be propagated through code, using dependent types. In contrast, our approach relies just on polymorphism, and does not require dependent types for writing subsidiary recursions. (*Subreci*, for reasoning about such recursions, of course does use dependent types).

Uustalu and Vene developed a categorical view of a recursion scheme allowing one level of subsidiary recursion, and illustrated it in Haskell with an artificial example [25]. In contrast, our scheme allows arbitrary finite nestings of recursion, and we illustrate it in Coq with realistic examples. It seems that generalizing the carriers of algebras to functors is the critical step enabling such examples.

Mendler-style recursion. Mendler introduced the idea of using universal abstraction to support compositional termination checking [16]. He proposed a functor-generic recursor of type $\forall X. (\forall R. (R \rightarrow X) \rightarrow F\ R \rightarrow X) \rightarrow \mu\ F \rightarrow X$. We have applied this idea to the constructor of the type *Mu* (Section 4.1). Previous work explored the categorical perspective on Mendler-style recursion, and showed how to reduce it to basic catamorphisms (i.e., structural recursion) [24]. Another considered its use with negative type schemes [1]. Previous work from our group showed how to derive inductive datatypes in Cedille using extensions of the Mendler encoding [9, 10]. Here, we do not derive inductive types, but rather a terminating recursion scheme for existing datatypes.

8 Conclusion

We have seen a derivation in Coq of a scheme for terminating subsidiary recursion, where recursions may be nested and outer recursive calls may be made on the results of inner recursions. We saw examples invoking the *span* function as a subsidiary recursion, for functions *wordsBy* and run-length encoding. We also looked briefly at the extension of this interface to support subsidiary induction, with example lemmas about *span*, and the decoding correctness theorem for run-length encoding. There are many other interesting examples we can develop in Coq with this interface, including natural-number division, which may invoke subtraction as a subsidiary recursion. Another example is Harper’s regular-expression matcher, which previous work showed can be implemented in Cedille using a form of nested recursion that is subsumed by subsidiary recursion [20]. We may also attempt to extend the recursion universe further, to allow other forms of recursion like divide-and-conquer, where some (necessarily limited) ability to recurse on values built using constructors is required.

References

- 1 Ki Yung Ahn and Tim Sheard. A hierarchy of mendler style recursion combinators: Taming inductive datatypes with negative occurrences. In *Proceedings of the 16th ACM SIGPLAN International Conference on Functional Programming, ICFP '11*, pages 234–246, New York, NY, USA, 2011. ACM.
- 2 Gilles Barthe, Maria João Frade, Eduardo Giménez, Luís Pinto, and Tarmo Uustalu. Type-based termination of recursive definitions. *Math. Struct. Comput. Sci.*, 14(1):97–141, 2004. URL: <https://doi.org/10.1017/S0960129503004122>, doi:10.1017/S0960129503004122.
- 3 Frédéric Blanqui. Inductive types in the calculus of algebraic constructions. *Fundam. Informaticae*, 65(1-2):61–86, 2005. URL: <http://content.iospress.com/articles/fundamenta-informaticae/fi65-1-2-04>.
- 4 Ana Bove, Alexander Krauss, and Matthieu Sozeau. Partiality and recursion in interactive theorem provers - an overview. *Mathematical Structures in Computer Science*, 26(1):38–88, 2016. URL: <https://doi.org/10.1017/S0960129514000115>, doi:10.1017/S0960129514000115.
- 5 Robin Cockett and Dwight Spencer. Strong categorical datatypes I. In R. A. G. Seely, editor, *International Meeting on Category Theory 1991*, Canadian Mathematical Society Proceedings. AMS, 1992.
- 6 Thierry Coquand and Christine Paulin. Inductively defined types. In Per Martin-Löf and Grigori Mints, editors, *COLOG-88, International Conference on Computer Logic, Tallinn, USSR, December 1988, Proceedings*, volume 417 of *Lecture Notes in Computer Science*, pages 50–66. Springer, 1988. URL: https://doi.org/10.1007/3-540-52335-9_47, doi:10.1007/3-540-52335-9_47.
- 7 Leonardo de Moura and Sebastian Ullrich. The lean 4 theorem prover and programming language. In André Platzer and Geoff Sutcliffe, editors, *Automated Deduction - CADE 28 - 28th International Conference on Automated Deduction, Virtual Event, July 12-15, 2021, Proceedings*, volume 12699 of *Lecture Notes in Computer Science*, pages 625–635. Springer, 2021. URL: https://doi.org/10.1007/978-3-030-79876-5_37, doi:10.1007/978-3-030-79876-5_37.
- 8 The Agda development team. *Agda*, 2021. Version 2.6.2.1. URL: <https://agda.readthedocs.io/en/v2.6.2.1/>.
- 9 Denis Firsov, Richard Blair, and Aaron Stump. Efficient mendler-style lambda-encodings in cedille. In Jeremy Avigad and Assia Mahboubi, editors, *Interactive Theorem Proving - 9th International Conference, ITP 2018, Held as Part of the Federated Logic Conference, FloC 2018, Oxford, UK, July 9-12, 2018, Proceedings*, volume 10895 of *Lecture Notes in Computer Science*, pages 235–252. Springer, 2018.
- 10 Denis Firsov and Aaron Stump. Generic derivation of induction for impredicative encodings in cedille. In June Andronick and Amy P. Felty, editors, *Proceedings of the 7th ACM SIGPLAN International Conference on Certified Programs and Proofs, CPP 2018, Los Angeles, CA, USA, January 8-9, 2018*, pages 215–227. ACM, 2018.
- 11 Tatsuya Hagino. *A Categorical Programming Language*. PhD thesis, University of Edinburgh, 1987.
- 12 Alexander Krauss. *Defining Recursive Functions in Isabelle/HOL*. URL: <https://isabelle.in.tum.de/doc/functions.pdf>.
- 13 Alexander Krauss. Partial and nested recursive function definitions in higher-order logic. *J. Autom. Reasoning*, 44(4):303–336, 2010. URL: <https://doi.org/10.1007/s10817-009-9157-2>, doi:10.1007/s10817-009-9157-2.
- 14 The Coq development team. *The Coq proof assistant reference manual*. LogiCal Project, 2021. Version 8.13.2. URL: <http://coq.inria.fr>.
- 15 Conor McBride. Elimination with a motive. In Paul Callaghan, Zhaohui Luo, James McKinna, and Robert Pollack, editors, *Types for Proofs and Programs, International Workshop, TYPES 2000, Durham, UK, December 8-12, 2000, Selected Papers*, volume 2277 of *Lecture Notes*

- 480 in *Computer Science*, pages 197–216. Springer, 2000. URL: [https://doi.org/10.1007/](https://doi.org/10.1007/3-540-45842-5_13)
481 3-540-45842-5_13, doi:10.1007/3-540-45842-5_13.
- 482 16 N. P. Mendler. Inductive types and type constraints in the second-order lambda calculus.
483 *Annals of Pure and Applied Logic*, 51(1):159 – 172, 1991.
- 484 17 Wolfgang Naraschewski and Tobias Nipkow. Isabelle/hol, 2020. URL: [http://www.cl.cam.](http://www.cl.cam.ac.uk/research/hvg/Isabelle/)
485 ac.uk/research/hvg/Isabelle/.
- 486 18 Scott Owens and Konrad Slind. Adapting functional programs to higher order logic. *Higher-*
487 *Order and Symbolic Computation*, 21(4):377–409, 2008. URL: [https://doi.org/10.1007/](https://doi.org/10.1007/s10990-008-9038-0)
488 s10990-008-9038-0, doi:10.1007/s10990-008-9038-0.
- 489 19 David Salomon and Giovanni Motta. *Handbook of Data Compression*. Springer, 2009.
- 490 20 Aaron Stump, Christopher Jenkins, Stephan Spahn, and Colin McDonald. Strong func-
491 tional pearl: Harper’s regular-expression matcher in cedille. *Proc. ACM Program. Lang.*,
492 4(ICFP):122:1–122:25, 2020. URL: [https://doi.org/10.1145/](https://doi.org/10.1145/3409004)
493 3409004.
- 494 21 Wouter Swierstra. Data types à la carte. *J. Funct. Program.*, 18(4):423–436, 2008. URL:
495 <https://doi.org/10.1017/S0956796808006758>, doi:10.1017/S0956796808006758.
- 496 22 Dmitriy Traytel, Andrei Popescu, and Jasmin Christian Blanchette. Foundational, composi-
497 tional (co)datatypes for higher-order logic: Category theory applied to theorem proving. In
498 *Proceedings of the 27th Annual IEEE Symposium on Logic in Computer Science, LICS 2012,*
499 *Dubrovnik, Croatia, June 25-28, 2012*, pages 596–605. IEEE Computer Society, 2012. URL:
500 <https://doi.org/10.1109/LICS.2012.75>, doi:10.1109/LICS.2012.75.
- 501 23 D. A. Turner. Elementary Strong Functional Programming. In *Proceedings of the First*
502 *International Symposium on Functional Programming Languages in Education*, FPLE ’95,
503 page 1–13, Berlin, Heidelberg, 1995. Springer-Verlag.
- 504 24 Tarmo Uustalu and Varmo Vene. Mendler-style inductive types, categorically. *Nordic J. of*
505 *Computing*, 6(3):343–361, September 1999.
- 506 25 Tarmo Uustalu and Varmo Vene. The Recursion Scheme from the Cofree Recursive Comonad.
507 *Electron. Notes Theor. Comput. Sci.*, 229(5):135–157, 2011. URL: [https://doi.org/10.1016/](https://doi.org/10.1016/j.entcs.2011.02.020)
508 j.entcs.2011.02.020, doi:10.1016/j.entcs.2011.02.020.